

```

*****
*           Smart NiCd Battery Charger           *
*****
* fosc=12.8MHz+5%by internal oscillator (tcycle=0.3125us)
*****
$noList
$include "fr908QT1.asm" ;frame file
$List

*I/O
out      equ 0 ;prtA,output for transistor
grn_LED  equ 4 ;prtA,output green LED
red_LED  equ 5 ;prtA,output red LED

*EQUATES
MAX_CNT  equ $E100 ;couner value for 16 hours
          ; (16h=57600sec->$E100)

*VARIABLES
          org    RAM
cnt       rmb    2    ;counter for Hours

*INITIALIZATION
          org    ROM
init      rsp                ;reset $ff-->SP
          bset   0,config1   ;COP disable
          mov    #%01000000,config2 ;set internal Osc
          lda    $ffc0       ;set fosc tolerance < 5%
          sta    $0038
          clr    cnt
          clr    cnt+1
          clrH
          clrX
          clrA
          clr    prtA
          mov    #%00110001,ddrA ;set I/O prtA
          mov    #%00110110,TSC  ;TOFInt disable,
          ;stop&clear Timer,set prescaler=64
          mov    #$c3,TmodH     ;set TOF period=1sec
          mov    #$50,TmodL     ; by $c350-->TMOD
          bclr   5,TSC          ;start TIMER
*****
Main      bset   red_LED,prtA   ; set red LED ON
          bset   out,prtA      ; set transistor ON
          brclr  TOF,TSC,*      ; wait for TOF=1
          bclr   TOF,TSC       ; reset TOF
          LDHX  cnt            ;cnt-->H:X
          aix   #1T            ;cnt+1
          STHX  cnt            ;H:X-->cnt
          CPHX  #MAX_CNT
          blo   Main           ;cnt<MAX_CNT?
*****
          bclr   out,prtA      ; set transistor OFF
          bclr   red_LED,prtA  ; set red LED OFF
          bset   grn_LED,prtA  ; set green LED ON
          bra   *              ; stop
*****
          org    $fffe ;Reset Vector
          fdb    init

.end
$noList

```